


CompuPro™

MDRIVE/H

TECHNICAL MANUAL

197C



MDRIVE/H™
TECHNICAL MANUAL

**512 KBYTE OR 2 MBYTE DISK EMULATOR
FOR THE IEEE 696/S-100 BUS**

MDrive/H Technical Manual ASY-197C
Copyright 1986 Viasyn Corporation
Hayward, CA 94545

Part No: 8261-0061A
Filename: MDRIVEH.MAN

DISCLAIMER - Viasyn Corporation makes no representations or warranties with respect to the contents hereof and specifically disclaims any implied warranties of merchantability or fitness for any particular purpose. Further, VIASYN reserves the right to revise this publication and to make any changes from time to time in the content hereof without obligation of VIASYN to notify any person of such revision or changes.

All rights reserved. No part of this publication may be reproduced or transmitted in any form, or by any means, without the written permission of VIASYN. Printed and assembled in the United States of America.

Registered Trademark: MDRIVE; Viasyn Corporation.

Trademarks: MDrive/H, the CompuPro logo; Viasyn Corporation.

Contents

Specifications	i
About the MDrive/H	
Configurations	1
Installing the MDrive/H	
Basic Installation	2
Switch Settings	3
Sample Switch Settings	4
Using the MDrive/H	5
Locating RAM ICs by Address and Byte	6
Programming Considerations	
Board Addressing	7
Data Reads and Writes	7
Sample Program	7
Schematic Diagrams	14
Component Layout	18
Warranty Information	19

Specifications

Timing	Meets all IEEE 696/S-100 specifications
Speed	RAM chip access time 150 ns.
Port Addresses	DIP switch selectable on any 2 port boundary. 0C6h and 0C7h standard in CompuPro operating systems
Logical Size	Up to eight MDrive/H boards can co-exist at the same two port addresses
Storage Capacity	512 Kbytes or 2 Mbytes per board (unformatted)
Memory Space Used	None
Power Consumption	Less than 8 watts typical, 10 watts maximum

About the MDrive/H

CompuPro helped pioneer the MDRIVE® concept for microcomputers. The MDRIVE concept is to use RAM memory to emulate a disk drive. Thus, the convenience and familiarity of a disk drive was preserved, but the drive now operates at RAM speeds, which is considerably faster than a mechanical disk drive. The original MDRIVE was implemented as a software product which used extended address memory beyond the first 64K under CP/M 80.

The MDrive/H is a "hardware" MDRIVE, extending this concept to all operating systems, without taking up precious memory space.

All of CompuPro's standard operating systems contain built in support for MDrive/H.

The MDrive/H board system is made up of one to eight boards, providing between 512 Kbytes and 16 Mbytes of total storage.

Configurations

The MDrive/H board is built in both a 512 Kbyte and a 2 Mbyte version. There are three differences between the versions: the number of RAM chips, the PAL installed at U17, and the PAL installed at U10. The following table depicts the differences.

<u>Board Size</u>	<u>U17</u>	<u>U10</u>	<u>RAM</u>
512 Kbytes	PAL # 420	PAL # 421	16 chips; Rows 6 & 7 filled
2 Mbyte	PAL # 287	PAL # 292	64 chips; All RAM sockets filled


NOTE: The number on PALs U10 and U17 and the settings of switch S1 (as described under Switch Settings) must match the number of RAM chips present.

Installing the MDrive/H Board

Basic Installation

Step 1. Unpack the MDrive/H Board.

Along with the board, you will find two card extractors in the plastic bag.



Step 2. Install Card Extractors.

1. Hold the board so the component side is toward you.
2. Insert the peg on the card extractor into the hole in the right corner of the board. Fold the extractor over the board's edge until the extractor's hole snaps over the peg.

NOTE: Make sure the long edge of the extractor is along the top edge of the board.

3. Repeat for the left extractor.

Step 3. Check Switch and Jumper Settings.

For standard switch settings for a CompuPro operating system check the operating system Installation Guide. Otherwise, refer to Switch Settings in this manual.

Step 4. Insert the MDrive/H into the S-100 Bus.

The power to the system must be off. Place the board into a slot towards the front of the enclosure. The edge connector is offset, so the MDrive/H will fit only one way. Push down GENTLY until the board is firmly installed.

Switch Settings

S1-1 (Switch 1 paddle 1) through S1-7 - Port Address Select

Switch 1 paddles 1-7 are used to select the two port addresses where the MDrive/H board resides. Up to 16 Mbytes of 2 Mbyte MDrive/H boards or up to 4 Mbytes of 512 Kbyte MDrive/H boards can exist at the same port addresses, only the cascade address (S1-8 through S1-10) needs to be different for each board.

The port address is set in a binary fashion with each paddle representing an address bit. An "ON" paddle represents a binary "zero" and an "OFF" paddle represents a binary "one". Paddle 1 is the MSB and corresponds with A7 and paddle 7 is the LSB and corresponds to A1. Since a pair of ports is addressed, A0 is not needed for selection.

The standard port addresses for the MDrive/H board are 0C6h and 0C7h.

S1-8 through S1-10 - Cascade Address Select

S1-8 through S1-10 select the cascade address for each board in the system. There are eight possible cascade addresses selected by the three paddles; paddle 8 is the MSB and paddle 10 is the LSB.


To set the cascade address, determine how many MDrive/H boards there are in the system, and set each of them according to the following table:

Board Number	Paddle Number		
	8	9	10
1st	ON	ON	ON
2nd	ON	ON	OFF
3rd	ON	OFF	ON
4th	ON	OFF	OFF
5th	OFF	ON	ON
6th	OFF	ON	OFF
7th	OFF	OFF	ON
8th	OFF	OFF	OFF


You may combine a single 2 Mbyte MDrive/H board with up to four 512 Kbyte MDrive/H boards. If you do combine boards, set the 2 Mbyte board at address 1 and the 512 Kbyte boards at addresses 5, 6, 7, and 8.

Sample Switch Settings

Four 2 Mbyte MDrive/H boards addressed for CompuPro software



One 2 Mbyte MDrive/H board and three 512 Kbyte MDrive/H boards addressed for CompuPro software



Using the MDrive/H

Most CompuPro software automatically formats the MDrive/H board(s) on BOOT and prints a message telling you how much MDrive/H storage it thinks is available (in Kbytes).

The operating system may be set up to use some of the MDrive/H as a hard disk cache. The remainder of the MDrive/H appears in a CompuPro system as drive M:. Once the system is booted, you may treat it just as if it were a disk drive.

All data will be lost in Drive M: when power is turned off, so be sure to transfer important work back to permanent media before turning off power. It's a good idea to do this periodically anyway, to protect against a power failure. The operating system handles writing its data to the hard disk, so you don't have to worry about losing cache data at power off.

The SHOW command determines the size of a drive in Kbytes. Check this number to see if the size of Drive M: seems correct. The size displayed by SHOW depends on the number and capacity of the boards in the system and on the amount of MDrive/H used as a hard disk cache and thus not available as a drive. If the size seems incorrect, power down the system and check your switch settings. If the size is correct, all is well.

Locating RAM ICs by Address and Byte

For an illustration of the board with row and column labels, refer to the Component Layout at the end of this manual.

The 256K DRAM chips are arranged on the board in rows of 8 chips. The rightmost chip in each row corresponds to Bit 0 of a memory byte and the leftmost chip corresponds to Bit 7.

In a 2 Mbyte board, eight rows of chips are filled. The lowest 256 Kbytes addressed by the board are in Row 0 (the top row) and the highest 256 Kbytes are in Row 7 (the bottom row).

NOTE: Row 0 Bit 0 is below Row 7 Bit 0 on the board.

In a 512 Kbyte board only two rows of chips are filled. The lower 256 Kbytes are in Row 6 and the higher 256 Kbytes are in Row 7.

Programming Considerations

The MDrive/H board consists of 2 Mbytes or 512 Kbytes of dynamic RAM, a dynamic RAM controller circuit and a method for addressing the data in the RAM array.

Board Addressing

The MDrive/H looks like two I/O ports to the system bus; it takes up no memory space. One I/O port is used to load a starting address into the board, and the other is used to read and write data.

<u>I/O Address</u>	<u>Function</u>	<u>Type</u>
Base Address + 0	Data Read/Write Port	R/W
Base Address + 1	Counter Address Load Port	Write Only

The starting address is loaded into a counter. The counter is loaded with three successive writes, MSB first. The counter has 24 bits, three more than are needed for 2 Mbytes and five more than are needed for 512 Kbytes. These extra bits are compared with the settings of S1-8 through S1-10 to determine which board should send or receive the data. Thus an array of eight 2 Mbyte boards seems to the software like a contiguous 16 Mbytes instead of eight separate chunks of 2 Megabytes each. A similar situation is true for 512 Kbyte boards. This makes the programming task quite a bit easier, as well as using a minimum number of port addresses.

Data Reads and Writes

When the read/write port is addressed, the counter increments the RAM address at the end of each read or write. Bytes are then transferred in sequence without the need to send a new address for each byte.

Sample Program

The following example of programming for the MDrive/H board consists of drivers that can be assembled into an existing BIOS. We are not saying that this is the best way to program for the MDrive/H. It is a sample program only.

This code expects the MDRIVE/H board to be addressed at the following I/O ports:

0C6H - Data
0C7H - Address

The disk parameter blocks must be set for the capacity of MDrive/H in the system.

```

1: ; M - D R I V E / H
2: ;
3: ;
4: ;
5: ;
6: HBASE: EQU 0C6H
7: HDATA: EQU HBASE
8: HADDR: EQU HBASE+1
9:
10: MEMDRV: EQU 'M' ;SET AS DRIVE "M"
11: FSRDAT: EQU 6 ;READ DATA
12: FSWRT: EQU 5 ;WRITE DATA
13:
14: MRTRY: EQU 10 ;MAX RETRY COUNT
15: ;
16: MDRIVE: JMP NOTHING ;COLD START
17: JMP NOTHING ;WARM START
18: JMP NOTHING ;CONSTAT
19: JMP NOTHING ;CONIN
20: JMP NOTHING ;CONOUT
21: JMP NOTHING ;LIST
22: JMP NOTHING ;PUSCH
23: JMP NOTHING ;READER
24: JMP HOME ;HOME
25: JMP SELDSK ;SELECT DISK
26: JMP SETTRK ;SET TRACK
27: JMP SETSEC ;SET SECTOR
28: JMP SETDMA ;SET DMA
29: JMP READ ;READ
30: JMP WRITE ;WRITE
31: JMP NOTHING ;LIST STATUS
32: JMP SECTRN ;SECTOR TRANSLATE
33: ;
34: ;NOTHING
35: ;
36: ;Dummy routine for non-implemented routines
37: ;
38: NOTHING:
39: RET
40: ;
41: HMDSIB: EQU 2048/128 ;SECTORS IN MEMORY DRIVE DATA BLOCK
42: HMDDSM: EQU ((512-4)*8)/HMDSIB
43: ;

```

```

44: DPBHMD: ;SET FOR 1 BOARD, 8 SPT
45: DB 128 ;MDRIVE TYPE FOR MFORM, DO NOT CHANGE
46: DW 8 ;SECTORS PER TRACK, THIS AND DSM ...
47: ;...MUST BE INCREASED FOR MORE BOARDS
48: DB 4,15,1 ;BSH, BLM, EXM
49: DW HMDDSM-1,128-1 ;DSM , NUMBER OF DIRECTORY ENTRIES
50: DB 11000000B,00000000B
51: ;RESERVED DIR BLOCKS, MUST MATCH ...
52: ; ... NUMBER OF DIRECTORY ENTRIES
53: DW 0 ;CHECK SUM
54: DW 4 ;NUMBER OF RESERVED TRACKS FOR
55: ;PARITY CHECK THIS IS ALWAYS 4
56: ;
57: HMDDPH:
58: DW 0 ;SECTOR TRANSLATE TABLE
59: DW 0,0,0 ;SCRATCH
60: DW BUFDIR ;DIRECTORY BUFFER
61: DW DPBHMD+1 ;DPB BLOCK
62: DW 0 ;CHECKSUM VECTOR
63: DW ALV ;ALLOCATION VECTOR
64: ;
65: ;
66: ; S E L E C T D I S K D R I V E
67: ;
68: ; Select the disk drive for subsequent disk transfers and return
69: ; the address of the DPH for the memory drive.
70: ;
71: ; ENTRY -- C = disk selection value
72: ; DE and I = 0 must determine disk type
73: ; = 1 type already determined (only needed for floppies)
74: ;
75: ; EXIT -- HL = DPH address of the memory drive.
76: ; IF memory drive not selected, control is passed to BIOS
77: ;
78: SELDSK: XRA A ;CLEAR A
79: STA SELFLG ;CLEAR MEMORY DRIVE FLAG
80: MOV A,C ;
81: CPI MEMDRV-'A' ;IS MEMORY DRIVE SELECTED
82: BIOSSEL: JNZ NOTHING ;F NOT GO BACK TO BIOS
83: MVI A,OFFH
84: STA SELFLG ;SET MEMORY DRIVE AS SELECTED
85: LXI H,HMDDPH ;GET DPH ADDRESS
86: RET ;AND DONE
87: ;
88: ; H O M E
89: ; Set current track to 0
90: ;
91: HOME: LXI H,0
92: SHLD MTRACK
93: BIOSHOM: JMP NOTHING ;ON TO BIOS
94: ;

```



```

95: ; SET TRACK
96: ;
97: ; Set track number. The track number is save for later use during
98: ; a disk transfer. The track is saved whether or not the memory disk
99: ; is currently selected.
100: ;
101: ; ENTRY -- BC = Track number.
102: ;
103: SETTRK: LXI H,0
104: DAD B
105: SHLD MTRACK
106: BIOSTRK:JMP NOTHING
107: ;
108: ; SET SECTOR
109: ;
110: ; Set sector number. The sector number is save for later use during
111: ; a disk transfer. The sector is saved whether or not the memory
112: ; disk is currently selected.
113: ;
114: ; ENTRY -- BC = Sector number.
115: ;
116: SETSEC: MOV A,C
117: STA MSECTOR
118: BIOSSEC:JMP NOTHING
119: ;
120: ; SET DMA ADDRESS
121: ;
122: ; Set the address for the next DMA transfer.
123: ; The DMA address is saved for later use during a disk transfer.
124: ; The address is saved whether or not the memory disk is
125: ; currently selected.
126: ;
127: ; ENTRY -- BC = Disk memory address.
128: ;
129: SETDMA: MOV L,C
130: MOV H,B
131: SHLD DMAADR
132: BIOSDMA:JMP NOTHING
133: ;
134: ; SECTOR TRANSLATE
135: ;
136: ; Translate the sector number from logical to physical.
137: ;
138: ; ENTRY DE = 0, no translation required.
139: ; DE = Translation table address.
140: ; BC = Number to translate.
141: ; EXIT HL = Translated sector.
142: ;
143: SECTRN:
144: PUSH B
145: POP H ;SET UP HL FOR EXIT

```

```

146: MOV A,D
147: ORA E
148: RZ ;NO TRANSLATION NEEDED
149: JMP NOTHING ;LET BIOS TRANSLATE
150: ;
151: ; R E A D
152: ;
153: ; Read the currently selected track,sector from the currently
154: ; selected drive.
155: ;
156: READ: LDA SELFLG ;IS MDRIVE CURRENTLY
157: ORA A ;SELECTED DISK
158: BIOSRED:JZ NOTHING
159: MVI A,F$RDAT ;MUST WANT THE MEMORY DRIVE
160: STA MCMD ;SET COMMAND AS "READ"
161: JMP HMDPNL ;AND GO PROCESS
162: ;
163: ; W R I T E
164: ;
165: ; Write to the currently selected drive at the current track
166: ; and sector.
167: ;
168: WRITE: LDA SELFLG
169: ORA A ;IS MEMORY DRIVE SELECTED
170: BIOSWRT:JZ NOTHING
171: MVI A,F$WRT ;SET COMMAND AS WRITE
172: STA MCMD
173: ; JMP HMDPNL ;AND GO PROCESS
174: ;
175: ; HMDPNL -- Hard memory disk final command processing
176: ; ENTRY -- MCMD = COMMAND
177: ; MTRACK = TRACK
178: ; MSECTOR = SECTOR
179: HMDPNL:
180: MVI A,MRTRY ;SET RETRY COUNT
181: STA RTRY
182: HMFNL1: LHL MTRACK ;GET TRACK
183: DAD H ;*2
184: DAD H ;*4
185: DAD H ;*8
186: DAD H ;*16
187: DAD H ;*32
188: DAD H ;*64
189: DAD H ;*128
190: LDA MSECTOR ;GET SECTOR
191: CALL SETMDRV ;SET MDRIVE/H TO ADDRESS IN A:HL
192: MVI C,128 ;128 BYTE SECTORS
193: MOV B,C ; ALSO INTO B FOR CHECKSUM
194: LHL DMAADR ;DIRECTLY TO CPM BUFFER
195: LDA MCMD ;GET COMMAND
196: CPI F$RDAT ;

```

```


197:      JNZ  HDINTO      ;IF WRITE COMMAND JUMP
198: ;
199:      JMP  HDOTOP
200: ;
201: ; MDRIVE/H READ
202: ;
203: ; Get a 128 byte block from MDRiveH board, address is already selected
204: ; C is number of bytes to move
205: ; HL is DMA buffer to put data
206: HDOTOP: IN    HDATA      ;GET DATA BYTE
207:      MOV  M,A          ;PUT TO DMA BLOCK
208:      ADD  B            ;ADD UP CHECKSUM
209:      MOV  B,A          ; AND SAVE
210:      INX  H            ;UP MEMORY POINTER
211:      DCR  C            ;COUNT THIS BYTE
212:      JNZ  HDOTOP      ;AND DO AGAIN
213:      CALL CHECKSUM      ;SET MDRIVE/E TO CHECKSUM
214:      IN   HDATA        ;GET CHECKSUM
215:      SUB  B            ; 0 IF NO ERROR
216:      RZ                ; THEN RETURN
217:      LDA  RTRY         ;GET RETRY COUNT
218:      DCR  A            ;COUNT THIS RETRY
219:      STA  RTRY         ;REPLACE WITH NEW COUNT
220:      JNZ  HMFNLI      ;LOOP IF MORE RETRIES
221:      ORI  1            ;SHOW HARD ERROR
222:      RET                ;RETURN WHEN DONE
223: ;
224: ; MDRIVE/H WRITE
225: ;
226: ; Put a block of data into MDrive/H board, address is already selected
227: ; C is number of bytes to move
228: ; HL is DMA buffer to get data from
229: ;
230: HDINTO:
231:      MOV  A,M          ;GET BYTE FROM BUFFER
232:      OUT  HDATA        ;TO MDRIVE BOARD
233:      ADD  B            ;ADD UP CHECKSUM
234:      MOV  B,A          ; AND SAVE
235:      INX  H            ;BUMP MEMORY POINTER
236:      DCR  C            ;COUNT THIS MOVE
237:      JNZ  HDINTO      ;JUMP IF MORE DATA
238:      CALL CHECKSUM      ;SELECT CHECKSUM LOCATION
239:      MOV  A,B          ;WRITE CHECKSUM TO MEMORY DISK
240:      OUT  HDATA        ;WRITE CHECKSUM TO MEMORY DISK
241:      XRA  A            ;AND SHOW NO ERROR
242:      RET                ;RETURN
243: ;
244: ;
245: ; SET CHECKSUM
246: ;

```


```

247: ; Find location of checksum for track in MTRACK
248: ; and the sector in MSECTOR
249: CHECKSUM:LDA  MSECTOR      ;GET SECTOR
250:      LHL  MTRACK          ;GET TRACK
251: ;
252: ;SET MDRIVE
253: ;
254: ;Set MDRIVE/H board to address in A,HL
255: ;
256: ;SETHDRV:
257:      OUT  HADDR          ;TO MDRIVE/H BOARD
258:      MOV  A,B
259:      OUT  HADDR
260:      MOV  A,L            ;UPPER BIT OF L PICKS TRACK 0-3
261:      OUT  HADDR          ;REMAINDER OF BYTE PICKS SECTOR AND BYTE
262:      RET
263: ;
264: MDRLNG: EQU  $-MDRIVE      ;LENGTH OF MDRIVE BIOS
265: SELFLG: DB   0            ;DISK SELECT FLAG
266: RTRY:   DB   0            ;RETRY COUNTER
267: MCMD:   DB   0
268: MSECTOR:DB  0
269: MTRACK: DW   0
270: DMAADR: DW   0            ;ADDRESS OF DMA BUFFER
271: BUFDIR: DS  128          ;DIRECTORY BUFFER FOR CPM
272: ALV:   DS   (((HDDSM*8)+7)/8)+1 ;
273: ;



```




Board Size	U17	RAM
512 Kbytes	PAL # 420	16 chips Rows 6 & 7 filled
2 Mbyte	PAL # 287	64 chips; All RAM sockets filled



Board Size	U10	RAM
512 Kbytes	PAL # 421	16 chips; Rows 6 & 7 filled
2 Mbyte	PAL # 292	64 chips; All RAM sockets filled





Component Layout

LIMITED WARRANTY

Viasyn Corporation warrants this computer product to be in good working order for a period of ninety days from the date of purchase by the original end user. Should this product fail to be in good working order at any time during this warranty period, VIASYN will, at its option, repair or replace the item at no additional charge except as set forth below. Repair parts and replacement products will be furnished on an exchange basis and will be either reconditioned or new. All replaced parts and products become the property of VIASYN. This limited warranty does not include service to repair damage to the product resulting from accident, disaster, misuse, abuse or unauthorized modification of the product.

If you need assistance, or suspect an equipment failure, always contact your System Center or dealer first. System Center technicians are trained to provide prompt diagnosis and repair of equipment failures. If you are not satisfied by the actions taken by your System Center or dealer, please call VIASYN at (415) 786-0909 to obtain a Return Material Authorization (RMA) number, or write to VIASYN at 26538 Danti Court, Hayward, CA, 94545-3999, Attn: RMA. Be sure to include a copy of the original bill of sale to establish a purchase date. If the product is delivered by mail or common carrier, you agree to insure the product or assume the risk of loss or damage in transit, to prepay shipping charges to the warranty service location and to use the original shipping container or equivalent. Be sure to mark the RMA number on the outside of the shipping container or delivery may be refused. Contact your System Center/dealer or write to VIASYN at the above address for further information.

All expressed and implied warranties for this product, including the warranties of merchantability and fitness for a particular purpose, are limited in duration to the above listed periods from the date of purchase and no warranties, either expressed or implied will apply after this period.

If this product is not in good working order as warranted above, your sole remedy shall be repair or replacement as provided above. In no event shall VIASYN be liable to you for any damages, including any lost profits, lost savings or other incidental or consequential damages arising out of the use of or inability to use such product, even if VIASYN or a System Center/dealer has been advised of the possibility of such damages, or for any claim by any other party.

If this product is out of warranty, please call or write the VIASYN RMA department to obtain a quotation for factory service. If this product was sold as a system by VIASYN, it may eligible and you may elect to purchase on site/depot maintenance from SPERRY. Contact your System Center/Dealer, your nearest SPERRY office or VIASYN for more details.

If you have purchased a SPERRY service and maintenance agreement, the following two paragraphs also apply:

If VIASYN or its service contractor fails after repeated attempts to perform any of its obligations set forth in this agreement, VIASYN's or its service contractor's entire liability and VIASYN's customer's sole and exclusive remedy for claims related to or arising out of this agreement for any cause and regardless of the form of action, whether in contract or tort, including negligence and strict liability, shall be Viasyn's customer's actual, direct damages such as would be provable in a court of law, but not to exceed the cost of the item of equipment involved.

In no event shall VIASYN or its service contractor be liable for any incidental, indirect, special or consequential damages, including but not limited to loss of use, revenue or profit, even if VIASYN or its service contractor has been advised, knew or should have known of the possibility of such damages; or damages caused by VIASYN's customer's failure to perform its obligations under this agreement; or claims, demands or actions against VIASYN's customer by any other party.

Viasyn Corporation
 26538 Danti Court
 Hayward, CA 94545-3999
 (415) 786-0909
 TWX 510-100-3288

EFFECTIVE 9/1/86. This warranty supersedes all previous warranties. All previous editions are obsolete.

LIMITED WARRANTY

Viasyn Corporation warrants this computer product to be in good working order for a period of ninety days from the date of purchase by the original end user. Should this product fail to be in good working order at any time during this warranty period, VIASYN will, at its option, repair or replace the item at no additional charge except as set forth below. Repair parts and replacement products will be furnished on an exchange basis and will be either reconditioned or new. All replaced parts and products become the property of VIASYN. This limited warranty does not include service to repair damage to the product resulting from accident, disaster, misuse, abuse or unauthorized modification of the product.

If you need assistance, or suspect an equipment failure, always contact your System Center or dealer first. System Center technicians are trained to provide prompt diagnosis and repair of equipment failures. If you are not satisfied by the actions taken by your System Center or dealer, please call VIASYN at (415) 786-0909 to obtain a Return Material Authorization (RMA) number, or write to VIASYN at 26538 Danti Court, Hayward, CA, 94545-3999, Attn: RMA. Be sure to include a copy of the original bill of sale to establish a purchase date. If the product is delivered by mail or common carrier, you agree to insure the product or assume the risk of loss or damage in transit, to prepay shipping charges to the warranty service location and to use the original shipping container or equivalent. Be sure to mark the RMA number on the outside of the shipping container or delivery may be refused. Contact your System Center/dealer or write to VIASYN at the above address for further information.

All expressed and implied warranties for this product, including the warranties of merchantability and fitness for a particular purpose, are limited in duration to the above listed periods from the date of purchase and no warranties, either expressed or implied will apply after this period.

If this product is not in good working order as warranted above, your sole remedy shall be repair or replacement as provided above. In no event shall VIASYN be liable to you for any damages, including any lost profits, lost savings or other incidental or consequential damages arising out of the use of or inability to use such product, even if VIASYN or a System Center/dealer has been advised of the possibility of such damages, or for any claim by any other party.

If this product is out of warranty, please call or write the VIASYN RMA department to obtain a quotation for factory service. If this product was sold as a system by VIASYN, it may be eligible and you may elect to purchase on site/depot maintenance from SPERRY. Contact your System Center/Dealer, your nearest SPERRY office or VIASYN for more details.

If you have purchased a SPERRY service and maintenance agreement, the following two paragraphs also apply:

If VIASYN or its service contractor fails after repeated attempts to perform any of its obligations set forth in this agreement, VIASYN's or its service contractor's entire liability and VIASYN's customer's sole and exclusive remedy for claims related to or arising out of this agreement for any cause and regardless of the form of action, whether in contract or tort, including negligence and strict liability, shall be Viasyn's customer's actual, direct damages such as would be provable in a court of law, but not to exceed the cost of the item of equipment involved.

In no event shall VIASYN or its service contractor be liable for any incidental, indirect, special or consequential damages, including but not limited to loss of use, revenue or profit, even if VIASYN or its service contractor has been advised, knew or should have known of the possibility of such damages; or damages caused by VIASYN's customer's failure to perform its obligations under this agreement; or claims, demands or actions against VIASYN's customer by any other party.

Viasyn Corporation
26538 Danti Court
Hayward, CA 94545-3999
(415) 786-0909
TWX 510-100-3288

EFFECTIVE 9/1/86. This warranty supersedes all previous warranties. All previous editions are obsolete.

Viaysn Corporation
26538 Dantl Court, Hayward, CA 94545-3999 (415) 786-0909

\$20.00 8261-0061A

42650C/1